

# T.D. 10 – Corrigé

## Les flags et les boucles

### Exercice 1

Donnez le résultat des additions suivantes, ainsi que le contenu des bits **N**, **Z**, **V** et **C** du registre d'état.

1. \$8054 + \$AB3D (opération en .W)  
= \$12B91 avec **N = 0**, **Z = 0**, **V = 1** et **C = 1**
2. \$7F + \$7A (opération en .B)  
= \$F9 avec **N = 1**, **Z = 0**, **V = 1** et **C = 0**

### Exercice 2

Réalisez une addition 64 bits en respectant les indications suivantes :

Entrées : **D1:D0** = Entier sur 64 bits (**D0** étant les 32 bits de poids faible).

**D3:D2** = Entier sur 64 bits (**D2** étant les 32 bits de poids faible).

Sortie : **D1:D0** = **D1:D0** + **D3:D2**

```
add.l    d2,d0      ; d2 + d0      -> d0, retenue -> X
addx.l   d3,d1      ; d3 + d1 + X -> d1, retenue -> X
```

### Exercice 3

Réalisez un programme qui détermine la valeur absolue d'un entier signé.

Entrée : **D0.L** = Entier signé sur 32 bits.

Sortie : **D0.L** = **|D0.L|**

```
abs                ; Mise à jour des flags Z et N en fonction de d0.
                  ; Si d0 est positif ou nul, alors N = 0.
                  ; Si d0 est négatif, alors N = 1.
tst.l    d0

                  ; Saut à quit si N = 0 (donc si d0 est positif).
bpl.s    quit

                  ; Sinon N = 1 (donc d0 est négatif).
                  ; Calcule le complément à 2 de d0.
neg.l    d0

quit            ; Sortie du programme.
```

**Exercice 4**

Réalisez un programme qui code le contenu de la mémoire, compris entre l'adresse \$2000 et \$3000, de la manière suivante :

Un mot de passe de 4 lettres est situé à l'adresse pointée par **A0**.

On réalise un OU exclusif entre la 1<sup>re</sup> lettre du mot passe et l'adresse \$2000.

On réalise un OU exclusif entre la 2<sup>e</sup> lettre du mot passe et l'adresse \$2001.

On réalise un OU exclusif entre la 3<sup>e</sup> lettre du mot passe et l'adresse \$2002.

On réalise un OU exclusif entre la 4<sup>e</sup> lettre du mot passe et l'adresse \$2003.

On réalise un OU exclusif entre la 1<sup>re</sup> lettre du mot passe et l'adresse \$2004.

On réalise un OU exclusif entre la 2<sup>e</sup> lettre du mot passe et l'adresse \$2005.

Et ainsi de suite jusqu'à l'adresse \$3000 incluse.

```
                ; Initialise a1 avec l'adresse de départ.
                lea    $2000,a1

                ; Initialise le compteur de boucle d7.
                ; ($1001 itérations avec l'utilisation de DBRA.)
                move.w  #$1000,d7

                ; Initialise d0 avec le mot de passe de 4 lettres (32 bits).
                move.l  (a0),d0

loop            ; Positionne la prochaine lettre du mot de passe à utiliser
                ; dans les 8 bits de poids faible de d0.
                rol.l   #8,d0

                ; Ou exclusif entre la donnée et la lettre du mot passe.
                ; (Incrémentation de a1 pour passer à l'adresse suivante.)
                eor.b   d0,(a1)+

                ; Reboucle tant que d7 est différent de -1.
                dbra    d7,loop
```